1.1 Some Basic Mathematical Models; Direction Fields

Maple Setup

We'll need the DEtools package to make DEplot available. (Recall that green, underlined text is a hyperlink that opens Help pages, other Maple worksheets, or Internet sites.) **DEplot** graphs direction fields with solution curves. End the statement with a colon to suppress printing the entire list of routines in the **DEtools** package. We'll also use the D operator for differentiating functions and diff for differentiating expressions.

```
> with(DEtools):
```

Problem 11

Draw a direction field for the given differential equation. Based on the direction field, determine the behavior of y as t approaches infinity. If this behavior depends on the initial value of y at $t = 0$, describe this dependency.

11. $y' = y(4 - y)$

First,

```
> with(DEtools):
```

Define the differential equation remembering that Maple requires us to use the full function expression $y(t)$, not y. We'll use the *operator notation* D(y) for the derivative function, then y' is entered as D(y)(t).

```
> ode := D(y)(t) = y(t)*(4-y(t));
```

$$ode := \mathrm{D}(y)(t) = y(t)\ (4 - y(t))$$

Let's go directly to graphing a direction field. The basic syntax for DEplot is

DEplot(*differential_equation, dependent_var, independent_var_range, dependent_var_range*)

(Enter "**? DEplot**" or click on the green link above for more information.)

Since the roots of the right side $y(4\text{-}y)$ are 0 & 4 and the right side is independent of t, we'll take y in [-2,6] and t in [-2,4], a shorter range suffices because of the independence.

```
> DEplot(ode, y(t), t=-2..4, y=-2..6);
```



The direction field shows the *equilibrium solutions*, $y = 0$ and $y = 4$. The arrow pattern tells us that, for $y_0 > 0$, solution curves will approach $y = 4$. For $y_0 < 0$, the curves will tend to negative infinity.

Let's regraph the direction field, adding several solution curves using y_0 = -0.1, 0, 0.5, 3, 4, and 6. Define the initial values as a list, enclosed in square brackets, of points $[0, y_0]$.

```
> InitialValues := [[0,-0.1], [0,0], [0,0.5], [0,3], [0,4], [0,
  6]];
```

$$InitialValues := [[0, -0.1], [0, 0], [0, 0.5], [0, 3], [0, 4], [0, 6]]$$

Make the new graph. (The default curves are yellow; we'll switch to black for visibility.)

```
> DEplot(ode, y(t), t=0..4, y=-2..6, InitialValues, linecolor=
  black);
```



>

The image above verifies our observations and shows how quickly *y* moves toward 4 and away from 0.